/*
 *  Licensed to the Apache Software Foundation (ASF) under one or more
 *  contributor license agreements.  See the NOTICE file distributed with
 *  this work for additional information regarding copyright ownership.
 *  The ASF licenses this file to You under the Apache License, Version 2.0
 *  (the "License"); you may not use this file except in compliance with
 *  the License.  You may obtain a copy of the License at
 *
 *      https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing, software
 *  distributed under the License is distributed on an "AS IS" BASIS,
 *  WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 *  See the License for the specific language governing permissions and
 *  limitations under the License.
 */

import java.time.Instant
import java.time.LocalDate
import java.time.ZoneOffset
import java.time.format.DateTimeFormatter

ext {
    isReproducibleBuild = System.getenv("SOURCE_DATE_EPOCH") != null
    buildInstant = java.util.Optional.ofNullable(System.getenv("SOURCE_DATE_EPOCH"))
            .filter(s -> !s.isEmpty())
            .map(Long::parseLong)
            .map(Instant::ofEpochSecond)
            .orElseGet(Instant::now)
    formattedBuildDate = System.getenv("SOURCE_DATE_EPOCH") ?
            DateTimeFormatter.ISO_INSTANT.format(buildInstant) :
            DateTimeFormatter.ISO_DATE.format(LocalDate.ofInstant(buildInstant as Instant, ZoneOffset.UTC))
    buildDate = (buildInstant as Instant).atZone(ZoneOffset.UTC) // for reproducible builds

    grailsVersion = projectVersion
    isCiBuild = System.getenv().get('CI') as Boolean
    configuredTestParallel = findProperty('maxTestParallel') as Integer ?: (isCiBuild ? 3 : Runtime.runtime.availableProcessors() * 3 / 4 as int ?: 1)
    excludeUnusedTransDeps = findProperty('excludeUnusedTransDeps')

    testProjectsStartWith = [
            'grails-test-suite',
            'grails-test-examples'
    ]
    testProjects = [ /* Will be populated by subprojects loop below */]
    docProjects = [ /* Will be populated by subprojects loop below */]
    cliProjects = [ /* Will be populated by subprojects loop below */]
    profileProjects = [ /* Will be populated by subprojects loop below */]
}

subprojects {

    for (String testPrefix : testProjectsStartWith) {
        if (name.startsWith(testPrefix)) {
            rootProject.ext['testProjects'] << name
        }
    }

    if (name.startsWith('grails-doc') || name.startsWith('grails-data-doc') || name.endsWith('-docs') || name.endsWith('-test-report')) {
        rootProject.ext['docProjects'] << name
    }

    if (name.startsWith('grails-wrapper')) {
        rootProject.ext['cliProjects'] << name
    }

    if (name.startsWith('grails-profiles')) {
        rootProject.ext['profileProjects'] << name
    }

    configurations.configureEach {
        resolutionStrategy {
            def cacheHours = isCiBuild || isReproducibleBuild ? 0 : 24
            cacheDynamicVersionsFor(cacheHours, 'hours')
            cacheChangingModulesFor(cacheHours, 'hours')
        }
    }
}

apply {
    from layout.projectDirectory.file('gradle/publish-root-config.gradle')
    from layout.projectDirectory.file('gradle/rat-root-config.gradle')
}

// For debugging the gradle task graph: Uncomment to show task dependencies when a task is run
//gradle.taskGraph.whenReady {taskGraph ->
//    logger.lifecycle("Found ${taskGraph.allTasks.size()} tasks.")
//    taskGraph.allTasks.forEach { task ->
//        logger.lifecycle("${task}")
//        task.dependsOn.forEach { dep ->
//            logger.lifecycle("\t- ${dep}")
//        }
//    }
//}